
Configuring Ecto repository stats
This guide covers how to configure the LiveDashboard to stats from your underlying database. At the moment, these stats can only be show for Ecto repositories running on Ecto.Adapters.Postgres.
Installing Ecto Stats
To enable the "Ecto Stats" functionality in your dashboard, you will need to do the three steps below:
	Add the ecto_psql_extras dependency
	Configure the dashboard
	(optional) Install custom extensions

Add the ecto_psql_extras dependency
In your mix.exs, add the following to your deps:
  {:ecto_psql_extras, "~> 0.2"},
Configure the dashboard
The next step is to configure the dashboard. Go to the live_dashboard call in your router and list all of your repositories:
live_dashboard "/dashboard", ecto_repos: [MyApp.Repo]
You want to list all repositories that connect to distinct databases. For example, if you have both MyApp.Repo and MyApp.RepoAnother but they connect to the same database, there is no benefit in listing both. Remember only Ecto repositories running on Ecto.Adapters.Postgres are currently supported.
If you want to disable the "Ecto Stats" option altogether, set ecto_repos: [].
Install custom extensions
Once the repository page is enabled, some of the queries (Calls and Outliers) require the pg_stat_statements extension installed. If you wish to access said functionality, you must install the extension first, otherwise an error will be shown.


  

Configuring metrics
This guide covers how to install and configure your LiveDashboard Metrics.
Installing metrics
To enable the "Metrics" functionality in your dashboard, you will need to do the three steps below:
	Add the telemetry dependencies
	Define your telemetry module
	Configure the dashboard

Add the telemetry dependencies
In your mix.exs, add the following to your deps:
  {:telemetry_poller, "~> 0.4"},
  {:telemetry_metrics, "~> 0.4"},
If you generated your Phoenix app in version v1.5+, these dependencies will already be installed. You can also skip the next section.
Define your telemetry module
In your Phoenix application, we recommend that you create a module to act as your telemetry supervision tree. Within this supervisor you can define your application's metrics and start your reporters.
The example below contains the child spec for a LiveDashboard reporter, as well as some metrics definitions for telemetry events emitted by Phoenix, Ecto, and the VM (via the :telemetry_poller package).
Create your Telemetry module in lib/my_app_web/telemetry.ex:
defmodule MyAppWeb.Telemetry do
  use Supervisor
  import Telemetry.Metrics

  def start_link(arg) do
    Supervisor.start_link(__MODULE__, arg, name: __MODULE__)
  end

  @impl true
  def init(_arg) do
    children = [
      # Telemetry poller will execute the given period measurements
      # every 10_000ms. Learn more here: https://hexdocs.pm/telemetry_metrics
      {:telemetry_poller, measurements: periodic_measurements(), period: 10_000}
      # Add reporters as children of your supervision tree.
      # {Telemetry.Metrics.ConsoleReporter, metrics: metrics()}
    ]

    Supervisor.init(children, strategy: :one_for_one)
  end

  def metrics do
    [
      # Phoenix Metrics
      summary("phoenix.endpoint.stop.duration",
        unit: {:native, :millisecond}
      ),
      summary("phoenix.router_dispatch.stop.duration",
        tags: [:route],
        unit: {:native, :millisecond}
      ),

      # Database Time Metrics
      summary("my_app.repo.query.total_time", unit: {:native, :millisecond}),
      summary("my_app.repo.query.decode_time", unit: {:native, :millisecond}),
      summary("my_app.repo.query.query_time", unit: {:native, :millisecond}),
      summary("my_app.repo.query.queue_time", unit: {:native, :millisecond}),
      summary("my_app.repo.query.idle_time", unit: {:native, :millisecond}),

      # VM Metrics
      summary("vm.memory.total", unit: {:byte, :kilobyte}),
      summary("vm.total_run_queue_lengths.total"),
      summary("vm.total_run_queue_lengths.cpu"),
      summary("vm.total_run_queue_lengths.io")
    ]
  end

  defp periodic_measurements do
    []
  end
end
Make sure to replace MyApp and my_app by your actual application name.
Then add to your main application's supervision tree (usually in lib/my_app/application.ex):
children = [
  MyApp.Repo,
  MyAppWeb.Telemetry,
  MyAppWeb.Endpoint,
  ...
]
Configure the dashboard
The last step now is to configure the dashboard. Go to the live_dashboard call in your router and add the following option:
live_dashboard "/dashboard", metrics: MyAppWeb.Telemetry
Now refresh the "/dashboard" page and the metrics functionality should be enabled. Each metric goes to a distinct group based on the metric name itself.
More about telemetry
Now that you have metrics up and running, you can begin exploring the rest of the telemetry ecosystem! Here are a few links to get you started:
	The Telemetry.Metrics
module documentation contains more information on:
	Metric types
	Breaking down metrics by tags
	VM Metrics
	Custom periodic polling


	For a deeper dive into Phoenix and Ecto metrics, see our
Telemetry Walkthrough.

	For more Elixir libraries using :telemetry, see
Libraries using Telemetry.


Configure Metrics
The LiveDashboard integrates with :telemetry converting each  Telemetry.Metrics to a beautiful, real-time chart.
The following table shows how Telemetry.Metrics metrics map to LiveDashboard charts:
	Telemetry.Metrics	Y-Axis Value(s)
	last_value	Always set to an absolute value
	counter	Always increased by 1
	sum	Always increased/decreased by an absolute value
	summary	Value/Min/Max/Avg
	distribution	(Coming Soon) recording measurements in individual buckets

Reporter options
Reporter options can be given to each metric as an option. For example:
counter("my_app.counter", reporter_options: [...])
The following reporter options are available to the dashboard:
	:group - configures the group the metric belongs to. By default the group is the first part of the name. For example, counter("my_app.counter") defaults to group "my_app"

	:prune_threshold - the maximum number of data
points. When the threshold is reached, the chart data will
be pruned by half. Default is 1_000.


Metrics history
Metrics history can also be enabled via a custom configuration.


  

Configuring metrics history
If you wish to populate metrics with history saved from telemetry or another data source,
modify the dashboard config to include a metrics_history key like so:
live_dashboard "/dashboard",
  metrics: MyAppWeb.Telemetry,
  metrics_history: {MyApp.MyStorage, :metrics_history, []}
where MyStorage is a module and :metrics_history is a function taking a single argument in this example, which will always be a metric.  The function must return a list, empty if there is no data, or a list of maps with :label, :measurement and :time keys in every map.  The function Phoenix.LiveDashboard.extract_datapoint_for_metric/4 will return a map in exactly this format (with optional time argument if you want to override the default of System.system_time(:microsecond)), or it may return nil in which case the data point should not be saved.
As an example, if you want history for all metrics, you can store history for those metrics in a circular buffer and emit recent telemetry when each client connects. LiveDashboard calls into your module for history for the metrics on that tab.  You would also need to add the module to your Application children, and initialize it with some or all of your metrics, such as from MyAppWeb.Telemetry.metrics/0. You could store the data in an ETS table or in Redis or the database, or anywhere else, but for this example we'll show using a GenServer and the circular_buffer library:
  defmodule MyApp.MyStorage do
    use GenServer

    @history_buffer_size 50

    def metrics_history(metric) do
      GenServer.call(__MODULE__, {:data, metric})
    end

    def start_link(args) do
      GenServer.start_link(__MODULE__, args, name: __MODULE__)
    end

    @impl true
    def init(metrics) do
      Process.flag(:trap_exit, true)

      metric_histories_map =
        metrics
        |> Enum.map(fn metric ->
          attach_handler(metric)
          {metric, CircularBuffer.new(@history_buffer_size)}
        end)
        |> Map.new()

      {:ok, metric_histories_map}
    end

    @impl true
    def terminate(_, metrics) do
      for metric <- metrics do
        :telemetry.detach({__MODULE__, metric, self()})
      end

      :ok
    end

    defp attach_handler(%{event_name: name_list} = metric) do
      :telemetry.attach(
        {__MODULE__, metric, self()},
        name_list,
        &__MODULE__.handle_event/4,
        metric
      )
    end

    def handle_event(_event_name, data, metadata, metric) do
      if data = Phoenix.LiveDashboard.extract_datapoint_for_metric(metric, data, metadata) do
        GenServer.cast(__MODULE__, {:telemetry_metric, data, metric})
      end
    end

    @impl true
    def handle_cast({:telemetry_metric, data, metric}, state) do
      {:noreply, update_in(state[metric], &CircularBuffer.insert(&1, data))}
    end

    @impl true
    def handle_call({:data, metric}, _from, state) do
      if history = state[metric] do
        {:reply, CircularBuffer.to_list(history), state}
      else
        {:reply, [], state}
      end
    end
  end


  

Configuring OS Data
This guide covers how to install and configure your LiveDashboard OS Data.
Enabling os_mon
The OS Data comes from the os_mon application, which ships as part of your Erlang distribution. You can start it by adding it to the extra applications section in your mix.exs:
  def application do
    [
      ...,
      extra_applications: [:logger, :runtime_tools, :os_mon]
    ]
  end
Configuring os_mon
See the Erlang docs for more information and os_mon configuration.


  

Configuring request logger
This guide covers how to install and configure your LiveDashboard request logger.
Installing request logger
Installing the request logger is straight-forward. Just add the following plug to your "lib/my_app_web/endpoint.ex", right before Plug.RequestId:
plug Phoenix.LiveDashboard.RequestLogger,
  param_key: "request_logger",
  cookie_key: "request_logger"
If your application is an API only, then you most likely don't use cookies, which means you can remove the "cookie_key" option.
Now go to your dashboard and the Request Logger will be enabled. Once you click it, you will have the option to either pass a parameter to stream requests logs or to enable/disable a cookie that streams requests logs.
Configuring request logger
There is no configuration for the request logger at the moment.

LiveDashboard provides real-time performance monitoring and debugging tools for Phoenix developers. It provides the following modules:
	Home - See general information about the system

	OS Data - See general information about OS, such as CPU, Memory and Disk usage

	Metrics - See how your application performs under different conditions by visualizing :telemetry events with real-time charts

	Request logging - See everything that was logged for certain requests

	Applications - See, filter, and search applications in the current node

	Processes - See, filter, and search processes in the current node

	Ports - See, filter, and search ports (responsible for I/O) in the current node

	Sockets - See, filter, and search sockets (responsible for tcp/udp) in the current node

	ETS - See, filter, and search ETS tables (in-memory storage) in the current node

	Ecto Stats - Shows index, table, and general usage about the underlying Ecto Repo storage


The dashboard also works across nodes. If your nodes are connected via Distributed Erlang, then you can access information from node B while accessing the dashboard on node A.
[image: screenshot]
Installation
To start using LiveDashboard, you will need three steps:
	Add the phoenix_live_dashboard dependency
	Configure LiveView
	Add dashboard access

1. Add the phoenix_live_dashboard dependency
Add the following to your mix.exs and run mix deps.get:
def deps do
  [
    {:phoenix_live_dashboard, "~> 0.1"}
  ]
end
2. Configure LiveView
The LiveDashboard is built on top of LiveView. If LiveView is already installed in your app, feel free to skip this section.
If you plan to use LiveView in your application in the future, we recommend you to follow the official installation instructions.
This guide only covers the minimum steps necessary for the LiveDashboard itself to run.
First, update your endpoint's configuration to include a signing salt. You can generate a signing salt by running mix phx.gen.secret 32 (note Phoenix v1.5+ apps already have this configuration):
# config/config.exs
config :my_app, MyAppWeb.Endpoint,
  live_view: [signing_salt: "SECRET_SALT"]
Then add the Phoenix.LiveView.Socket declaration to your endpoint:
socket "/live", Phoenix.LiveView.Socket
And you are good to go!
3. Add dashboard access for development-only usage
Once installed, update your router's configuration to forward requests to a LiveDashboard with a unique name of your choosing:
# lib/my_app_web/router.ex
use MyAppWeb, :router
import Phoenix.LiveDashboard.Router

...

if Mix.env() == :dev do
  scope "/" do
    pipe_through :browser
    live_dashboard "/dashboard"
  end
end
This is all. Run mix phx.server and access the "/dashboard" to configure the necessary modules.
Extra: Add dashboard access on all environments (including production)
If you want to use the LiveDashboard in production, you should put it behind some authentication and allow only admins to access it. If your application does not have an admins-only section yet, you can use Plug.BasicAuth to set up some basic authentication as long as you are also using SSL (which you should anyway):
# lib/my_app_web/router.ex
use MyAppWeb, :router
import Plug.BasicAuth
import Phoenix.LiveDashboard.Router

...

pipeline :admins_only do
  plug :basic_auth, username: "admin", password: "a very special secret"
end

scope "/" do
  pipe_through [:browser, :admins_only]
  live_dashboard "/dashboard"
end
If you are running your application behind a proxy or a webserver, you also have to make sure they are configured for allowing WebSocket upgrades. For example, here is an article on how to configure Nginx with Phoenix and WebSockets.
Finally, you will also want to configure your config/prod.exs and use your domain name under the check_origin configuration:
check_origin: ["//myapp.com"]
Then you should be good to go!

      Link to this function
    
    extract_datapoint_for_metric(metric, measurements, metadata, time \\ nil)


      
       
       View Source
     


  


  

      Specs

      

          extract_datapoint_for_metric(
  Telemetry.Metric.t(),
  map(),
  map(),
  pos_integer() | nil
) :: %{label: binary(), measurement: number(), time: pos_integer()} | nil


      


Extracts a datapoint for the given metric.
Receives a Telemetry.Metric as metric, alongside the measurements
and metadata from the Telemetry event, and an optional time and
returns an extracted datapoint or nil if the event is not part of
the metric.
Note that it is expected that the event name was already validated as
part of the metric.

  


        

      


  

    
Phoenix.LiveDashboard.PageBuilder behaviour
    



      
Page builder is the default mechanism for building custom dashboard pages.
Each dashboard page is a LiveView with additional callbacks for
customizing the menu appearance. One notable difference, however,
is that a page implements a render_page/1 callback, which must
return one or more page builder components, instead of a render/1
callback that returns ~L.
A simple and straight-forward example of a custom page is the
Phoenix.LiveDashboard.ETSPage that ships with the dashboard:
defmodule Phoenix.LiveDashboard.EtsPage do
  @moduledoc false
  use Phoenix.LiveDashboard.PageBuilder

  @impl true
  def menu_link(_, _) do
    {:ok, "ETS"}
  end

  @impl true
  def render_page(_assigns) do
    table(
      columns: columns(),
      id: :ets_table,
      row_attrs: &row_attrs/1,
      row_fetcher: &fetch_ets/2,
      rows_name: "tables",
      title: "ETS"
    )
  end

  defp fetch_ets(params, node) do
    %{search: search, sort_by: sort_by, sort_dir: sort_dir, limit: limit} = params

    # Here goes the code that goes through all ETS tables, searches
    # (if not nil), sorts, and limits them.
    #
    # It must return a tuple where the first element is list with
    # the current entries (up to limit) and an integer with the
    # total amount of entries.
    # ...
  end

  defp columns() do
    [
      %{
        field: :name,
        header: "Name or module",
      },
      %{
        field: :protection
      },
      %{
        field: :type
      },
      %{
        field: :size,
        cell_attrs: [class: "text-right"],
        sortable: :desc
      },
      %{
        field: :memory,
        format: &format_words/1,
        sortable: :desc
      },
      %{
        field: :owner,
        format: &encode_pid/1
      }
    ]
  end

  defp row_attrs(table) do
    [
      {"phx-click", "show_info"},
      {"phx-value-info", encode_ets(table[:id])},
      {"phx-page-loading", true}
    ]
  end
end
Once a page is defined, it must be declared in your live_dashboard
route as follows:
live_dashboard "/dashboard",
  additional_pages: [
    route_name: MyAppWeb.MyCustomPage
  ]
Or alternatively:
live_dashboard "/dashboard",
  additional_pages: [
    route_name: {MyAppWeb.MyCustomPage, some_option: ...}
  ]
The second argument of the tuple will be given to the init/1
callback. If not tuple is given, init/1 will receive an empty
list.
Components
A page can only have the components listed with this page.
At the moment, only nav_bar/1 and table/1 are supported.

    nav_bar(assigns)

  


    Renders a nav bar.






  
    table(assigns)

  


    Renders a table component.

    Callback invoked when a page is declared in the router.

    Callback invoked when a page is declared in the router.

      Link to this type
    
    capabilities()


      
       
       View Source
     


  


  

      Specs

      

          capabilities() :: %{
  applications: [atom()],
  modules: [atom()],
  processes: [atom()],
  dashboard_running?: boolean(),
  system_info: nil | binary()
}


      



  



  
    
      
      Link to this opaque
    
    component()


      
       
       View Source
     


      (opaque)

  


  

      Specs

      

          component()


      



  



  
    
      
      Link to this type
    
    requirements()


      
       
       View Source
     


  


  

      Specs

      

          requirements() :: [{:application | :process | :module, atom()}]


      



  



  
    
      
      Link to this type
    
    session()


      
       
       View Source
     


  


  

      Specs

      

          session() :: map()


      



  



  
    
      
      Link to this type
    
    unsigned_params()


      
       
       View Source
     


  


  

      Specs

      

          unsigned_params() :: map()

      Link to this function
    
    nav_bar(assigns)


      
       
       View Source
     


  


  

      Specs

      

          nav_bar(keyword()) :: component()


      


Renders a nav bar.
It can be rendered in any dashboard page via the render_page/1 function:
def render_page(assigns) do
  nav_bar(
    items: [
      phoenix_metrics: [
        name: "Phoenix Metrics",
        render: table(...)
      ],

      vm_metrics: [
        name: "VM Metrics",
        render: fn -> table(...expensive_parameters) end
      ]
    ]
  )
end
You can see it in use the Metrics and Ecto info pages.

  



  
    
      
      Link to this function
    
    table(assigns)

      Specs

      

          table(keyword()) :: component()


      


Renders a table component.
It can be rendered in any dashboard page via the render_page/1 function:
def render_page(assigns) do
  table(
    columns: columns(),
    id: @table_id,
    row_attrs: &row_attrs/1,
    row_fetcher: &fetch_applications/2,
    title: "Applications"
  )
end
You can see it in use the applications, processes, sockets pages and
many others.
Options
These are the options supported by the component:
	:id - Required. Because is a stateful Phoenix.LiveComponent an unique id is needed.

	:columns - Required. A Keyword list with the following keys:
	:field - Required. An identifier for the column.
	:header - Label to show in the current column. Default value is calculated from :field.
	:header_attrs - A list with HTML attributes for the column header. Default: [].
	:format - Function which receives the value and returns the cell information.
Default is the field value itself.
	:cell_attrs - A list with HTML attributes for the table cell. Default: [].
	:sortable - Either :asc or :desc with the default sorting. When set, the column
header is clickable and it fetches again rows with the new order. At least one column
should be sortable. Default: nil


	:row_fetcher - Required. A function which receives the params and the node and
returns a tuple with the rows and the total number:
(params(), node()) -> {list(), integer() | binary()}

	:rows_name - A string to name the representation of the rows.
Default is calculated from the current page.

	:default_sort_by - The default columnt to sort by to.
Defaults to the first sortable column.

	:title - The title of the table.
Default is calculated with the current page.

	:limit - A list of integers to limit the number of rows to show.
Default: [50, 100, 500, 1000, 5000]. May be set to false to disable the limit.

	:search - A boolean indicating if the search functionality is enabled.
Default: true.

	:hint - A textual hint to show close to the title. Default: nil.

      Link to this callback
    
    handle_event(event, unsigned_params, socket)


      
       
       View Source
     


      (optional)

  


  

      Specs

      

          handle_event(event :: binary(), unsigned_params(), socket :: Socket.t()) ::
  {:noreply, Socket.t()} | {:reply, map(), Socket.t()}


      



  



  
    
      
      Link to this callback
    
    handle_info(msg, socket)


      
       
       View Source
     


      (optional)

  


  

      Specs

      

          handle_info(msg :: term(), socket :: Socket.t()) :: {:noreply, Socket.t()}


      



  



  
    
      
      Link to this callback
    
    handle_params(unsigned_params, uri, socket)


      
       
       View Source
     


      (optional)

  


  

      Specs

      

          handle_params(unsigned_params(), uri :: String.t(), socket :: Socket.t()) ::
  {:noreply, Socket.t()}


      



  



  
    
      
      Link to this callback
    
    handle_refresh(socket)


      
       
       View Source
     


      (optional)

  


  

      Specs

      

          handle_refresh(socket :: Socket.t()) :: {:noreply, Socket.t()}


      



  



  
    
      
      Link to this callback
    
    init(term)


      
       
       View Source
     


  


  

      Specs

      

          init(term()) :: {:ok, session()} | {:ok, session(), requirements()}


      


Callback invoked when a page is declared in the router.
It receives the router options and it must return the
tuple {:ok, session, requirements}.
The page session will be serialized to the client and
received on mount.
The requirements is an optional keyword to detect the
state of the node.
The result of this detection will be passed as second
argument in the menu_link/2 callback.
The possible values are:
	:applications list of applications that are running or not.
	:modules list of modules that are loaded or not.
	:pids list of processes that alive or not.


  



  
    
      
      Link to this callback
    
    menu_link(session, capabilities)


      
       
       View Source
     


  


  

      Specs

      

          menu_link(session(), capabilities()) ::
  {:ok, String.t()}
  | {:disabled, String.t()}
  | {:disabled, String.t(), String.t()}
  | :skip


      


Callback invoked when a page is declared in the router.
It receives the session returned by the init/1 callback
and the capabilities of the current node.
The possible return values are:
	{:ok, text} when the link should be enable and text to be shown.

	{:disabled, text} when the link should be disable and text to be shown.

	{:disabled, text, more_info_url} similar to the previous one but
it also includes a link to provide more information to the user.

	:skip when the link should not be shown at all.



  



  
    
      
      Link to this callback
    
    mount(unsigned_params, session, socket)


      
       
       View Source
     


      (optional)

  


  

      Specs

      

          mount(unsigned_params(), session(), socket :: Socket.t()) ::
  {:ok, Socket.t()} | {:ok, Socket.t(), keyword()}


      



  



  
    
      
      Link to this callback
    
    render_page(assigns)


      
       
       View Source
     


  


  

      Specs

      

          render_page(assigns :: Socket.assigns()) :: component()


      



  


        

      


  

    
Phoenix.LiveDashboard.RequestLogger
    



      
A plug that enables request logging.
See our Request Logger guides for more information.

      




  

    
Phoenix.LiveDashboard.Router
    



      
Provides LiveView routing for LiveDashboard.

      Functions
    


  
    live_dashboard(path, opts \\ [])

  


    Defines a LiveDashboard route.

      Link to this macro
    
    live_dashboard(path, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Defines a LiveDashboard route.
It expects the path the dashboard will be mounted at
and a set of options.

  
  Options


	:live_socket_path - Configures the socket path. it must match
the socket "/live", Phoenix.LiveView.Socket in your endpoint.

	:csp_nonce_assign_key - an assign key to find the CSP nonce
value used for assets

	:ecto_repos - the repositories to show database information.
Currently only PSQL databases are supported

	:env_keys - Configures environment variables to display.
It is defined as a list of string keys. If not set, the environment
information will not be displayed.

	:metrics - Configures the module to retrieve metrics from.
It can be a module or a {module, function}. If nothing is
given, the metrics functionality will be disabled.

	:metrics_history - Configures a callback for retreiving metric history.
It must be an "MFA" tuple of  {Module, :function, arguments} such as
  metrics_history: {MyStorage, :metrics_history, []}
If not set, metrics will start out empty/blank and only display
data that occurs while the browser page is open.

	:request_logger_cookie_domain - Configures the domain the request_logger
cookie will be written to. It can be a string or :parent atom.
When a string is given, it will directly  set cookie domain to the given
value. When :parent is given, it will take the parent domain from current
endpoint host (if host is "www.acme.com" the cookie will be scoped on
"acme.com"). When not set, the cookie will be scoped to current domain.

	:allow_destructive_actions - When true, allow destructive actions directly
from the UI. Defaults to false. The following destructive actions are
available in the dashboard:
	"Kill process" - a "Kill process" button on the process modal
Note that custom pages given to "Additional pages" may support their own
destructive actions.


	:additional_pages - A keyword list of addictional pages



  
  Examples


defmodule MyAppWeb.Router do
  use Phoenix.Router
  import Phoenix.LiveDashboard.Router

  scope "/", MyAppWeb do
    pipe_through [:browser]
    live_dashboard "/dashboard",
      metrics: {MyAppWeb.Telemetry, :metrics},
      env_keys: ["APP_USER", "VERSION"],
      metrics_history: {MyStorage, :metrics_history, []},
      request_logger_cookie_domain: ".acme.com"
  end
end

